
Introduction

In this column, we would like to discuss a simple problem with a simple solution. If you’ve programmed in Smalltalk for long enough, you will inevitably encounter situations where the response to a message is a collection of some sort. If the responding object maintains an internal collection, it can either return this collection, return a copy, or provide some other mechanism that gives access to the individual elements. Most designers would argue that it’s not a good idea to return the original collection because it makes future extensions more difficult. Changing the type of collection returned, for example, could adversely impact users.



If this is accepted as a bit of “programming wisdom”, what are some acceptable solutions? Additionally, is this folklore part of more general wisdom that should be spelled out? Does it go all the way to a design pattern? We believe that there is an interesting counter-intuitive design pattern involved here. By counter-intuitive, we mean that it’s advice we use ourselves but would never give to someone else. 

A Concrete Example

Consider a club; e.g., a book club, a video club, a club for remote controlled planes, a water-skiing club, … If a club is supposed to keep track of its members, novice designers will undoubtedly provide get/set methods such as:



	members

		^members

	members: aCollection

		members := aCollection



Conventional wisdom suggests that the get method makes future extensions more difficult because it provides users with unnecessarily detailed information about the collection. Additionally, it permits user to add new members or remove members from the collection without the club being aware of the changes. What if the introduction of a member requires a new entry in an “unpaid dues list”? What if the deletion of a member requires that the member be moved to a “past members” list. The set method is even more dangerous because it permits wholesale replacements to the collection without any notification to the club. 



Could we get around the criticisms by writing more sophisticated methods like the following?



	members

		^members asArray

	members: aCollection

		self terminateOldMembers: members.

		members := aCollection.

		self initiateNewMembers: members



Perhaps! But no one ever writes code like this. The get method would probably end up making unnecessary collection copies 95% (or maybe even 100%) of the time; it’s unjustifiably inefficient. The set method, to contrast with the get method, just looks too complicated. 



The more typical solution is to leave the get method as it was but to expect the user to make the least assumptions about the type of collection used and to eliminate the set method in favor of add: and remove: methods (where needed). Following through with these suggestions, we could end up with methods such as the following:



	members

		"Use only do:, size, isEmpty, notEmpty."

		^members

	add: aClubMember

		self initiateMember: aClubMember

		members add: aClubMember.

	remove: aClubMember

		members remove: aClubMember.

		self terminateMember: aClubMember.



Unfortunately, we rarely document our get method expectations in enough detail. If we’re lucky, we might know that the result is an indexable collection. But shouldn’t we limit ourselves to a small subset of the indexable collection protocol? Surely, we don’t want destructive protocol to be used. With respect to the add: and remove: methods, wouldn’t it be better to merge add: and initiateMember: and also remove: and terminateMember:? That would better support the club object’s needs for other kinds of lists. The result might be methods such as the following:



	members

		"Use only do:, size, isEmpty, notEmpty."

		^members

	addMember: aClubMember

		… initiation code if needed …

		members add: aClubMember.

	removeMember: aClubMember

		… termination code if needed …

		members remove: aClubMember.



	bills

		^bills

	addBill: aBill

		…

	removeBill: aBill

		…



The next step undertaken by more experienced designers is to eliminate the get methods in favor of a sequencing facility. This way, there is no ambiguity about what should and should not be an allowable operation on a list. There is only one (a form of do:) and it is provided only through the club’s protocol.



	membersDo: aBlock

		members do: aBlock

	billsDo: aBlock

		bills do: aBlock

	

If the club object is redesigned at some future date so that the member’s list is changed from a set to a dictionary, no users are affected. The only problem with this is the rather restricted protocol it supports. If we want to support, select:, then we need to provide membersSelect: and billsSelect: methods. Normally, we don’t provide any of these extensions because it quickly gets out of hand.

Providing Less Restrictive Collection Protocols

The contribution of this column is to show how we can go one small step further. Rather than provide a sequencing facility or explicitly return a private collection, return an object that provides read-only access to this private collection without providing any significant additional overhead. Because the new object has a public protocol that can be studied, there is no ambiguity about what is and what is not an allowable message. The entire public protocol is available. 



There is no particular challenge to implementing such an object which we will call an iterator. It simply duplicates a subset of the collection protocol that is read-only. It is in effect simply a “protocol-restricting” object. Using iterators, our get method is rewritten as follows:



	members

		^Iterator on: members



Alternatively, with suitable extensions to collections and streams (see Listing 1), we could rewrite the method as follows:



	members

		^members asIterator



The advantage of iterators is that it supports all of the standard sequencing operations (even the baroque inject:into: message, as shown below) but does not support accidental attempts to modify the collection via, for example, an add: message.



	A legal operation:



	aClub members 

		inject: 0 into: [:total :member |

			total + member age].



	An illegal operation:



	aClub members add: #Junk



The role of iterators is not to prevent malicious users from modifying a collection but rather to allow friendly users to notice accidental modifications when the  intent is to provide them with read-only access. A secondary reason is to ensure that the protocol available is independent of the actual collection type.

The Iterator Implementation

A particular implementation of iterators is provided in Listing 2. One possible approach to implementing iterators is to route every single message through to the original collection. Messages like collect:, select:, and reject: would then return collections like the original private collection. Since the intent is to make users resilient to implementation changes, it seems more appropriate to return the same kind of collection no matter what the original happened to be. We decided to have these messages return ordered collections. We also debated whether we should support random access via a message such as “at: index” or “with: anotherCollection do: aBlock”. Ultimately, we decided to do it realizing that this would require a type conversion if the original collection was not indexable. Subsequent indexed access, however, would be efficient since the converted collection could replace the original in the iterator. In general, we didn’t follow through with all the useful indexable collection messages. 



At one point in our design, we had tried to support iterators on streams. After all, streams too understand the do: message. But this do: didn’t have the same semantics as the collection’s do: (it meant “continue from the current position” so that a second execution of do: would iterate over zero elements). We could handle this with a StreamIterator class by making the iterator’s do: sequence over the result of executing “privateStream shallowCopy; reset”. That way, the original stream was untouched. But to work, the new iterator class had to re-implement every sequencing operation from first principles. Rather than provide two separate classes: one that re-routes and one that re-implements, we generalized and discarded the re-routing version. 



In the end, we re-implemented everything except do: and size and we abandoned the goal of handling streams directly (we now convert to an indexable collection at iterator creation time). We contemplated going back to the re-routing approach for those messages that would work but felt it wasn’t worth the effort.



We also provided iterators with a number of novel sequencing operations that we subsequently retrofitted into our collection hierarchy. 



	groupDo: aBlock

	groupsDo: aBlock

	separateWith: aBlock

	indexedDo: aBlock

	when: booleanBlock collect: elementBlock

	all: booleanBlock

	any: booleanBlock



In our programming environment, message groupDo: gives us access to all the elements of a collection of known size via nice block parameter names. For example, it enables us to write



	| data |

	data := #(0 0 10 20).

	data groupDo: [:left :top :right :bottom |

		^Rectangle 

			origin: left@top 

			corner: right@bottom]



instead of writing



	| data |

	data := #(0 0 10 20).

	left := data first.

	top := data at: 2.

	right := data at: 3.

	bottom := data last



There is no limit to the number of block parameters. Message groupsDo: provides access to a collection of fixed size subcollections.



	#((0 0) (10 20) (30 50)) groupsDo: [:x :y |

		… do something with x@y …]



Message separateWith: (courtesy of Alan Knight) provides a mechanism for partitioning values into several groups. A dictionary is returned with one key per group. For example,



#(-200 -100 -5 -1 0 1 5 100 200)

	separateWith: [:value |

		value < 0 ifTrue: [#negative] ifFalse: [

		value = 0 ifTrue: [#zero] ifFalse: [

		#positive]]]



returns a dictionary with 3 keys (each with an associated ordered collection): 



	#negative ( OrderedCollection (-200 -100 -5 -1)

	#zero ( OrderedCollection (0)

	#positive ( OrderedCollection (1 5 100 200)



See Listing 3 for examples of the other sequencing operations.

Lessons to be Learned

One might be persuaded that the reason for adopting measures such as we described above is to better encapsulate our objects(to make less of an object’s internals visible and accessible to users. If that’s a worthwhile goal, our experienced programmers must be better at it than the novices and they must surely spend time ensuring that their objects are nicely-encapsulated. 



What we would like to argue here is that encapsulation is NOT what we strive for at all. In fact, objects give away their private parts wholeheartedly to any other object that requests them (TRY THIS FOR A PULL-QUOTE). If you need proof, walk up to any programmer and ask to see any one of his/her classes. At least 90% of the guts of an instance will be obtainable via simple query messages. Should you wish to be annoying, you can easily show how easy it is to “break” the instance by making destructive changes to the parts obtained. Of course, you’ll get the immediate obvious reaction(“you’re not supposed to do that” or “anyone can break a object if they really want to.”



So we really don’t try to hide an object’s private parts at all. Far from it, we actually strive to make all the useful ones publicly accessible. Something is fundamentally wrong if our advice to others is “make your objects black boxes and hide all of their internals as much as possible” when all of our own programming follows the adage “Oh, you need this information! Here, take this private part of mine. It’s exactly what you want.” So let’s digress for a moment to see where this eagerness to give away private information comes from.



The source of our eagerness comes from the Smalltalk (and Java) language itself. It encourages object sharing and makes object copying the explicit responsibility of the programmer. Assignments don’t copy (the variable on the left is made to reference the object on the right). Message parameters are not copied (the formal parameters reference the actual parameters). By contrast, C++ discourages object sharing (because it is difficult to explicitly release objects shared by multiple users). Consequently, it provides mechanisms like copy constructors to make copies transparently. These two wildly different philosophies must clearly have important and visible effects on the corresponding programming styles.



When an object is designed for use in Smalltalk, a contract is implied between the implementer and the user that specifies what is expected of both parties.  Since Smalltalk programmers have been implementing software for years without explicitly writing down the contract details, there must be some consensus about the typical expectations; otherwise, we wouldn’t be able to program together on the same project. What are those expectations?



Surely, we don’t program for malicious users. We expect the users of our objects to be cooperating with us. We expect them to be friends. Also, we program for efficiency. When a friend wants some information, we explicitly avoid making a copy, opting instead to provide the raw information directly. Since sharing is fundamental to Smalltalk semantics and copying is inherently expensive, we share unless that leads to a problem. Anthropomorphically speaking, objects readily respond to queries with “here’s a piece of myself(just don’t cut my part.” 



What programmers are really doing is designing objects that provide read-only access to private parts while retaining read-write access rights. Moreover, rather than provide policing that ensures users don’t modify these parts, we rely instead on the good judgment of the users to avoid making changes. It is the user’s responsibility to make explicit copies of pieces that ultimately will be changed. The problem is that we don’t have a mechanism that permits our friends to know when they have accidentally changed one of our private parts.



But that’s not quite all that good programmers do. They also strive to insulate users from changes to their implementation, not by hiding private parts, but by making the private parts as replaceable as possible. It’s not so much that we wish to restrict a user to using the fewest messages possible on the private parts. Rather, we wish to restrict users to a set of messages that many other objects understand. This will make it more likely that evolutionary changes to the parts will allow the same messages to be supported because it ensures that it’s not a one-of-a-kind object. It’s no guarantee, of course, but it is a useful technique for attempting to predict the future. Let’s call this process “making the parts maximally replaceable.”

A Design Pattern for Programming in a Language that Supports Sharing

In order to provide a useful design pattern, we clearly have to differentiate between object designs intended for friendly users and object designs intended for malicious (or potentially malicious) users. For malicious users, for example, it’s prudent to always return copies. But much more is required. Requests that ask objects to make changes to their own parts may need authentication if the parts can’t be subsequently recreated. The object must provide a firewall that can’t be thwarted. Programming for malicious contexts is a new area that we are only now starting to grapple with. But even these objects with firewalls are implemented with internal objects that are all friendly to each other. So it’s fair to say that the great majority (98%) of all objects designed are intended for friendly users.



We can summarize how to achieve good object designs with a simple design pattern that echoes the sharing philosophy encouraged by Smalltalk: “For friendly users, make your objects provide read-only access to publicly desirable parts using maximally replaceable protocol; read-only access makes the information efficiently accessible; maximally replaceable protocol makes future changes easier.” We view this as a design pattern rather than a programming idiom because it influences the semantics of all the query messages in the public protocol of our objects.



With this design pattern in hand, we can easily point out why the notion of iterators is a good thing. First, it is a facility that can be used to provide read-only access to an object’s collection. Second, it provides the user with a rich protocol for manipulating the resulting collection no matter what kind of collection it is, making the specific collection used highly replaceable.

Conclusions

We have argued that good designers don’t really encapsulate very well. Rather they strive to provide read-only access to object parts for maximum execution efficiency. They also strive to provide objects with maximal replacement potential to increase evolutionary possibilities. What this means is that a user’s access protocol on the results returned by an object is intended to be as generic as possible to permit as many options as possible when a replacement must be found.



We also argued that typical programming is not about guarding against malicious users, but about providing extensible disciplined access to friends. So it’s not about hiding parts but about making parts as visible as possible to objects that need them. It’s also about trying to ensure that as a user of some other object’s parts, we avoid accidentally modifying them.



We also suggested that the simple notion of an iterator could be used to provide more disciplined read-only access to an object’s collection and that it also served to make the collection more replaceable.



This suggests a capability that would be great in the next generation programming language; namely, the ability to make any object read-only to others but read-write to the owner.





Listing 1 Extensions to Streams and Collections.



instance conversion in class Stream

asIterator

	^Iterator on: self contents



instance conversion in class Collection

asIterator

	^Iterator on: self





Listing 2 The Iterator class.



class:									Iterator

superclass:							Object

instance variables:			privateCollection



comment							Can be used on objects that

											support do: and size.

class methods 



instance creation

on: aCollection

	^self new on: aCollection



instance methods



instance initialization

on: aCollection

	privateCollection := aCollection



sequencing/standard



collect: aBlock

	| result |

	result := OrderedCollection new.

	self do: [:object | 

		result add: (aBlock value: object)].

	^result



conform: aBlock

	self do: [:object | 

		(aBlock value: object) ifFalse: [^false]].

	^true



detect: aBlock

	^self 

		detect: aBlock 

		ifNone: [self error: 'Nothing found']



detect: aBlock ifNone: exceptionBlock

	self do: [:object | 

		(aBlock value: object) ifTrue: [^object]].

	^exceptionBlock value



do: aBlock

	privateCollection do: aBlock



inject: initialValue into: aBlock

	| result |

	result := initialValue.

	self do: [:object | 

		result := aBlock value: result value: object].

	^result



reject: aBlock

	| result |

	result := OrderedCollection new.

	self do: [:object | 

		(aBlock value: object) ifFalse: [

			result add: object]].

	^result



select: aBlock

	| result |

	result := OrderedCollection new.

	self do: [:object | 

		(aBlock value: object) ifTrue: [

			result add: object]].

	^result



with: aCollection do: aBlock

	self privateEnsureIndexable.

	privateCollection with: aCollection do: aBlock



sequencing/non-standard

all: booleanBlock

	"Answer true if every element satisfies the block test."

	^self conform: booleanBlock



any: booleanBlock

	"Answer true if one element satisfies the block test."

	self detect: booleanBlock ifNone: [^false].

	^true



groupDo: aBlock

	"Provides access to all elements with nice names."

	aBlock valueWithArguments: self asArray



groupsDo: aBlock

	"Provides access to all elements of subcollections with nice names."

	self do: [:array | 

		aBlock valueWithArguments: array]



indexedDo: aBlock

	"Provides access to all elements along with an index. Avoid using with:do: which requires that privateCollection be indexable."

	| index |

	index := 0.

	self do: [:element | 

		aBlock 

			value: element 

			value: (index := index + 1)]



separateWith: aBlock

	"Use the block's result to group the elements."

	| partitions partitionKey partitionCollection |

	partitions := Dictionary new.

	self do: [:object | 

		partitionKey := aBlock value: object.

		partitionCollection := partitions 

			at: partitionKey 

			ifAbsent: [

				partitions 

					at: partitionKey 

					put: OrderedCollection new].

		partitionCollection add: object].

	^partitions



when: booleanBlock collect: elementBlock

	| result |

	result := OrderedCollection new.

	self do: [:element |

		(booleanBlock value: element) ifTrue: [

			result add: (elementBlock value: element)]].

	^result



querying

size

	^privateCollection size



isEmpty

	^self size = 0



notEmpty

	^self isEmpty not



at: index

	self privateEnsureIndexable.

	^privateCollection at: index



includes: anObject

	^self includesEqual: anObject



includesEqual: anObject

	"Answer a Boolean which is true if anObject is equal to one of the receiver's elements and false otherwise."

	self do: [:object | 

		object = anObject ifTrue: [^true]].

	^false



includesIdentical: anObject

	"Answer a Boolean which is true if anObject is identical to one of the receiver's elements and false otherwise."

	self do: [:object | 

		object == anObject ifTrue: [^true]].

	^false



converting

asArray

	^self privateAddIndexedTo: 

		(Array new: self size)



asBag

	^self privateAddTo: Bag new



asByteArray

	privateCollection isString ifTrue: [

		^privateCollection asByteArray].

	^self privateAddIndexedTo: 

		(ByteArray new: self size)



asOrderedCollection

	^self privateAddTo: OrderedCollection new



asSet

	^self privateAddTo: Set new



asSortedCollection

	^self privateAddTo: SortedCollection new



asSortedCollection: aBlock

	^self privateAddTo: 

		(SortedCollection sortBlock: aBlock)



private

privateAddIndexedTo: collection

	self indexedDo: [:object :index | 

		collection at: index put: object]. 

	^collection



privateAddTo: collection

	self do: [:eachItem | 

		collection add: eachItem]. 

	^collection



privateEnsureIndexable

	privateCollection abrIsIndexableCollection 

		ifFalse: [privateCollection := self asArray]





Listing 3 Using the non-standard sequencing operations.



"The result is true."

(Iterator on: #(10 20 30 40)) 

	any: [:integer | integer > 25]



"The result is false."

(Iterator on: #(10 20 30 40)) 

	all: [:integer | integer > 25]



(Iterator on: #(10 20 100 200)) 

	groupDo: [:left :top :width :height | 

		^Rectangle 

			origin: left@top 

			extent: width@height].



(Iterator on: #((a b c) (10 20 30) ($a $b $c)))

	groupsDo: [:a :b :c | 

		Transcript cr; show: '1: ', a printString, 

			', 2: ', b printString, ', 3: ', c printString].



(Iterator on: #(a b c)) 

	indexedDo: [:object :index |

		Transcript cr; show: index printString, 

			': ', object printString].



((Iterator on: #(-200 -100 -5 -1 0 1 5 100 200)) 

	separateWith: [:value |

		value < -10 ifTrue: [#largeNegative] ifFalse: [

		value < 0 ifTrue: [#smallNegative] ifFalse: [

		value = 0 ifTrue: [#zero] ifFalse: [

		value < 10 ifTrue: [#smallPositive] ifFalse: [

		#largePositive]]]]]) inspect.



(Iterator on: #(10 20 30 40)) 

	when: [:integer | integer > 25] 

	collect: [:integer | integer + 1]

