Listing 1 The Iterator class.





class:									Iterator


superclass:							Object


instance variables:			privateCollection





comment							Can be used on objects that


											support do: and size.


class methods 





instance creation


on: aCollection


	^self new on: aCollection





examples





standardExamples


"


	(self on: #(one two three four)) 


		collect: [:string | string size]





	(self on: #(10 20 30 40)) 


		conform: [:string | string \\ 10 = 0]





	(self on: #('hello' 'bye' 'hi' 'so long')) 


		detect: [:string | string size < 4]


	(self on: #('hello' 'bye' 'hi' 'so long')) 


		detect: [:string | string size < 4] ifNone: [nil]


	(self on: #('hello' 'bye' 'hi' 'so long')) 


		detect: [:string | string size < 1] ifNone: [nil]





	(self on: #('once' 'upon' 'a' 'time'))


		inject: '' into: [:result :item | result, ' ', item]





	(self on: #('this' 'was' 'a' 'great' 'story')) 


		reject: [:string | string size < 4]





	(self on: #('this' 'was' 'a' 'great' 'story')) 


		select: [:string | string size < 4]





	(self on: #(10 20 30 40)) 


		with: #(1 2 3 4) do: [:integer1 :integer2 |


			Transcript cr; 


				show: (integer1 + integer2) printString]


"





nonStandardExamples


"


	(self on: #(10 20 30 40)) 


		any: [:integer | integer > 25]





	(self on: #(10 20 30 40)) 


		all: [:integer | integer > 25]





	(self on: #(a b c)) 


		groupDo: [:a :b :c | 


			Transcript cr; show: '1: ', a printString, ', 2: ', 


				b printString, ', 3: ', c printString].





	(self on: #((a b c) (10 20 30) ($a $b $c)))


		groupsDo: [:a :b :c | 


			Transcript cr; show: '1: ', a printString, ', 2: ', 


				b printString, ', 3: ', c printString].





	(self on: #(a b c)) 


		indexedDo: [:object :index |


			Transcript cr; show: index printString, 


				': ', object printString].





	((self on: #(-200 -100 -5 -1 0 1 5 100 200)) 


		separateWith: [:value |


			value < -10 ifTrue: [#largeNegative] ifFalse: [


			value < 0 ifTrue: [#smallNegative] ifFalse: [


			value = 0 ifTrue: [#zero] ifFalse: [


			value < 10 ifTrue: [#smallPositive] ifFalse: [


			#largePositive]]]]]) inspect.





	(self on: #(10 20 30 40)) 


		when: [:integer | integer > 25] 


		collect: [:integer | integer + 1]


"





conversionExamples


"


	| iterator |


	iterator := Iterator on: (Dictionary new 


		at: 'five' put: 5; at:  'six' put: 6; yourself).


	Transcript cr; cr; show: 'Dictionary test:'.


	Transcript cr; show: 'Set: ', 


		iterator asSet printString.


	Transcript cr; show: 'Bag: ', 


		iterator asBag printString.


	Transcript cr; show: 'Array: ', 


		iterator asArray printString.


	Transcript cr; show: 'OrderedCollection: ', 


		iterator asOrderedCollection printString.


	Transcript cr; show: 'SortedCollection: ', 


		(iterator asSortedCollection: [:string1 :string2 | 


				string1 size <= string2 size]) printString.





	| stream iterator |


	stream := ReadStream on: #(5 6). 


	stream next. 'should still include 5'.


	iterator := Iterator on: stream.


	Transcript cr; cr; show: 'Stream test:'.


	Transcript cr; show: 'Set: ', 


		iterator asSet printString.


	Transcript cr; show: 'Bag: ', 


		iterator asBag printString.


	Transcript cr; show: 'Array: ', 


		iterator asArray printString.


	Transcript cr; show: 'OrderedCollection: ', 


		iterator asOrderedCollection printString.


	Transcript cr; show: 'SortedCollection: ', 


		(iterator asSortedCollection: [:string1 :string2 | 


			string1 size <= string2 size]) printString.


"





queryingExamples


"


		| iterator |


		iterator := Iterator on: (Dictionary new 


			at: 'five' put: 'cinqo'; 


			at:  'six' put: 'siete'; 


			yourself).


		Transcript cr; show: 'Empty (no): ', 


			iterator isEmpty printString.


		Transcript cr; show: 'Not Empty (yes): ', 


			iterator notEmpty printString.


		Transcript cr; show: 'Size: ', 


			iterator size printString.


		Transcript cr; show: 'At 1: ', 


			(iterator at: 1) printString.


		Transcript cr; show: 'At 2: ', 


			(iterator at: 2) printString.


		Transcript cr; show: 'Includes (yes): ', 


			(iterator includes: 'siete') printString.


		Transcript cr; show: 'Includes Identical (yes): ', 


			(iterator includesIdentical: 


				(iterator at: 2)) printString.


		Transcript cr; show: 'Includes Identical (no): ', 


			(iterator includesIdentical: 


				'siete' shallowCopy) printString.


		Transcript cr; show: 'Includes Equal (yes) ', 


			(iterator includesEqual: 'siete') printString.





"





instance methods





instance initialization


on: aCollection


	privateCollection := aCollection





sequencing/standard





collect: aBlock


	| result |


	result := OrderedCollection new.


	self do: [:object | 


		result add: (aBlock value: object)].


	^result





conform: aBlock


	self do: [:object | 


		(aBlock value: object) ifFalse: [^false]].


	^true





detect: aBlock


	^self 


		detect: aBlock 


		ifNone: [self error: 'Nothing found']





detect: aBlock ifNone: exceptionBlock


	self do: [:object | 


		(aBlock value: object) ifTrue: [^object]].


	^exceptionBlock value





do: aBlock


	privateCollection do: aBlock





inject: initialValue into: aBlock


	| result |


	result := initialValue.


	self do: [:object | 


		result := aBlock value: result value: object].


	^result





reject: aBlock


	| result |


	result := OrderedCollection new.


	self do: [:object | 


		(aBlock value: object) ifFalse: [


			result add: object]].


	^result





select: aBlock


	| result |


	result := OrderedCollection new.


	self do: [:object | 


		(aBlock value: object) ifTrue: [


			result add: object]].


	^result





with: aCollection do: aBlock


	self privateEnsureIndexable.


	privateCollection with: aCollection do: aBlock





sequencing/non-standard


all: booleanBlock


	"Answer true if every element satisfies the block test."


	^self conform: booleanBlock





any: booleanBlock


	"Answer true if one element satisfies the block test."


	self detect: booleanBlock ifNone: [^false].


	^true





groupDo: aBlock


	"Provides access to all elements with nice names."


	aBlock valueWithArguments: self asArray





groupsDo: aBlock


	"Provides access to all elements of subcollections with nice names."


	self do: [:array | 


		aBlock valueWithArguments: array]





indexedDo: aBlock


	"Provides access to all elements along with an index. Avoid using with:do: which requires that privateCollection be indexable."


	| index |


	index := 0.


	self do: [:element | 


		aBlock 


			value: element 


			value: (index := index + 1)]





separateWith: aBlock


	"Use the block's result to group the elements."


	| partitions partitionKey partitionCollection |


	partitions := Dictionary new.


	self do: [:object | 


		partitionKey := aBlock value: object.


		partitionCollection := partitions 


			at: partitionKey 


			ifAbsent: [


				partitions 


					at: partitionKey 


					put: OrderedCollection new].


		partitionCollection add: object].


	^partitions





when: booleanBlock collect: elementBlock


	| result |


	result := OrderedCollection new.


	self do: [:element |


		(booleanBlock value: element) ifTrue: [


			result add: (elementBlock value: element)]].


	^result





querying


size


	^privateCollection size





isEmpty


	^self size = 0





notEmpty


	^self isEmpty not





at: index


	self privateEnsureIndexable.


	^privateCollection at: index





includes: anObject


	^self includesEqual: anObject





includesEqual: anObject


	"Answer a Boolean which is true if anObject is equal to one of the receiver's elements and false otherwise."


	self do: [:object | 


		object = anObject ifTrue: [^true]].


	^false





includesIdentical: anObject


	"Answer a Boolean which is true if anObject is identical to one of the receiver's elements and false otherwise."


	self do: [:object | 


		object == anObject ifTrue: [^true]].


	^false





converting


asArray


	^self privateAddIndexedTo: 


		(Array new: self size)





asBag


	^self privateAddTo: Bag new





asByteArray


	privateCollection isString ifTrue: [


		^privateCollection asByteArray].


	^self privateAddIndexedTo: 


		(ByteArray new: self size)





asOrderedCollection


	^self privateAddTo: OrderedCollection new





asSet


	^self privateAddTo: Set new





asSortedCollection


	^self privateAddTo: SortedCollection new





asSortedCollection: aBlock


	^self privateAddTo: 


		(SortedCollection sortBlock: aBlock)





private


privateAddIndexedTo: collection


	self indexedDo: [:object :index | 


		collection at: index put: object]. 


	^collection





privateAddTo: collection


	self do: [:eachItem | 


		collection add: eachItem]. 


	^collection





privateEnsureIndexable


	privateCollection abrIsIndexableCollection 


		ifFalse: [privateCollection := self asArray]








Listing 2 Extensions to Collection.





instance method





conversion


asIterator


	^Iterator on: self








Listing 3 Extensions to Stream.





instance method





conversion


asIterator


	^Iterator on: self contents





